<commit_before>SUMMARY = "Lightweight media scanner meant to be used in not-so-powerful devices"
AUTHOR = "ProFUSION"
HOMEPAGE = "http://lms.garage.maemo.org/"
SECTION = "libs/multimedia"
LICENSE = "LGPLv2.1+"
LIC_FILES_CHKSUM = "file://COPYING;md5=a6f89e2100d9b6cdffcea4f398e37343"
DEPENDS = "sqlite3 flac virtual/libiconv"

PR = "r2"

SRC_URI = "http://git.profusion.mobi/cgit.cgi/lightmediascanner.git/snapshot/release_${PV}.tar.bz2 \
"
SRC_URI[md5sum] = "f423376a70f8f321af69b12563b176fe"
SRC_URI[sha256sum] = "302a7f6cc355467cd20332f4e02c8b87ba6c0c7a6818a6a987e007aace19b41e"

S = "${WORKDIR}/release_${PV}"

PACKAGECONFIG ??= ""
PACKAGECONFIG[ogg] = "--enable-ogg,--disable-ogg,libvorbis"

inherit autotools pkgconfig

do_install_append() {
    install -d ${D}/${bindir}/
    install -m 755 ${S}/src/bin/.libs/test  ${D}/${bindir}/test-lms
}

PACKAGES =+ "${PN}-test"

FILES_${PN}-test = "${bindir}/test-lms"

FILES_${PN}-dbg += "${libdir}/${PN}/plugins/.debug"
FILES_${PN}-staticdev += "${libdir}/${PN}/plugins/*.a"

# otherwise fails with ERROR: could not add conversion charset 'UTF-16BE': Invalid argument
RDEPENDS_${PN}-test_append_libc-glibc = " glibc-gconv-utf-16"

# png.so jpeg.so id3.so are provided also by imlib2-loaders
PRIVATE_LIBS_${PN} = "video-dummy.so png.so jpeg.so id3.so pls.so audio-dummy.so rm.so ogg.so dummy.so m3u.so flac.so asf.so"<commit_msg>lightmediascanner: fix build for B!=S<commit_after>SUMMARY = "Lightweight media scanner meant to be used in not-so-powerful devices"
AUTHOR = "ProFUSION"
HOMEPAGE = "http://lms.garage.maemo.org/"
SECTION = "libs/multimedia"
LICENSE = "LGPLv2.1+"
LIC_FILES_CHKSUM = "file://COPYING;md5=a6f89e2100d9b6cdffcea4f398e37343"
DEPENDS = "sqlite3 flac virtual/libiconv"

PR = "r2"

SRC_URI = "http://git.profusion.mobi/cgit.cgi/lightmediascanner.git/snapshot/release_${PV}.tar.bz2 \
"
SRC_URI[md5sum] = "f423376a70f8f321af69b12563b176fe"
SRC_URI[sha256sum] = "302a7f6cc355467cd20332f4e02c8b87ba6c0c7a6818a6a987e007aace19b41e"

S = "${WORKDIR}/release_${PV}"

PACKAGECONFIG ??= ""
PACKAGECONFIG[ogg] = "--enable-ogg,--disable-ogg,libvorbis"

inherit autotools pkgconfig

do_install_append() {
    install -d ${D}/${bindir}/
    install -m 755 ${B}/src/bin/.libs/test  ${D}/${bindir}/test-lms
}

PACKAGES =+ "${PN}-test"

FILES_${PN}-test = "${bindir}/test-lms"

FILES_${PN}-dbg += "${libdir}/${PN}/plugins/.debug"
FILES_${PN}-staticdev += "${libdir}/${PN}/plugins/*.a"

# otherwise fails with ERROR: could not add conversion charset 'UTF-16BE': Invalid argument
RDEPENDS_${PN}-test_append_libc-glibc = " glibc-gconv-utf-16"

# png.so jpeg.so id3.so are provided also by imlib2-loaders
PRIVATE_LIBS_${PN} = "video-dummy.so png.so jpeg.so id3.so pls.so audio-dummy.so rm.so ogg.so dummy.so m3u.so flac.so asf.so"
{'fraction_non_alphanumeric': 0.16894812680115273, 'fraction_numerical': 0.07276657060518732, 'mean_word_length': 5.157427937915743, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5397697', 'n_tokens_mistral': 1302, 'n_tokens_neox': 1180, 'n_words': 189}